# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xclipboard
pkgver=1.1.3
pkgrel=1
pkgdesc="X clipboard manager"
arch=(i686 x86_64)
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxaw' 'libxmu' 'libxt' 'libx11')
makedepends=('xorg-util-macros')
source=(https://xorg.freedesktop.org/archive/individual/app/xclipboard-${pkgver}.tar.bz2)
sha256sums=('9c0e0a1f4f1d49f73c6cf3ccd27b5ca0e18afc63679f1a1ee466b1d82446ebb5')

build() {
  cd "${srcdir}/xclipboard-${pkgver}"
  ./configure --prefix=/usr
  make
}

package(){
  cd "${srcdir}/xclipboard-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
